

DataFlex Entwickler Tag 2017

Custom controls

Some fancy components that you
could be using too!

- Web Scheduler
- Questionnaire
- PDF Viewer
- Image Zoom

Web Scheduler

- Calendar control
- Based on dHtmlx Scheduler
- View and manage “event” data

Web Scheduler - highlights

- Multiple views
 - Day, Week, Month, Unit, Timeline
- Flexible & customizable
 - Various themes
 - Custom styling, texts, date formats, etc.
- Also works on mobile devices!
- Easy to use

Loading calendar Data



```
Procedure OnLoadEvents tWebDhxEvent[] ByRef aEvents String sMode Date dStart Date dEnd
  tWebDhxEvent currentEvent
  Integer iEventPtr
  DateTime dtFrom dtTo
  String sKey

  Move 0 to iEventPtr

  Get psSessionKey of oSessionManager to sKey

  Send Clear of oReservations_DD
  Move dStart to Reservations.FromDate
  Move dEnd to Reservations.ToDate
  Move sKey to Reservations.SessionKey

  Send Request_Read of oReservations_DD GT Reservations.File_Number 3 // find by FromDate, ID

  // Find all reservations for our date range and add them to our list of events
  While (Found)
    If ((Reservations.FromDate >= dStart) and (Reservations.ToDate <= dEnd) and ((Trim(Reservations.SessionKey) = sKey) or (Reservations.SessionKey = "DEMO"))) Begin

      // Add reservation to dataset
      Get MergeDateTime Reservations.FromDate Reservations.FromTime to dtFrom
      Get MergeDateTime Reservations.ToDate Reservations.ToTime to dtTo

      Move Reservations.Title to currentEvent.text
      Move dtFrom to currentEvent.start_date
      Move dtTo to currentEvent.end_date
      Move Reservations.ID to currentEvent.id
      Move Reservations.Color to currentEvent.color
      Move Reservations.RoomId to currentEvent.section_id
      Move Reservations.RoomId to currentEvent.unit_id

      Move currentEvent to aEvents[iEventPtr]

      Increment iEventPtr
    End

    Send Locate_Next of oReservations_DD
  Loop
End_Procedure
```



Augment scheduler events



```
Procedure OnEditEvent tWebDhxEvent tEvent
    Send PopupReservationDialog of oReservationPopup Self tEvent.id tEvent.section_id tEvent.start_date tEvent.end_date True
End_Procedure
```

```
Procedure OnCreateEvent tWebDhxEvent tEvent
    Send PopupReservationDialog of oReservationPopup Self tEvent.id tEvent.section_id tEvent.start_date tEvent.end_date True
End_Procedure
```

```
Procedure OnEventChange tWebDhxEvent tBefore tWebDhxEvent tAfter
    Send PopupReservationDialog of oReservationPopup Self tAfter.id tAfter.section_id tAfter.start_date tAfter.end_date True
End_Procedure
```

```
Procedure OnDeleteEvent tWebDhxEvent tEvent
End_Procedure
```

```
Procedure OnCloseModalDialog Handle hoModalDialog
    Forward Send OnCloseModalDialog hoModalDialog
```

```
    // An alternative to a full reload is to use the AddEvent and RemoveEvent procedures in the scheduler package
    Send Reload of oScheduler
End_Procedure
```

Questionnaire

- Form creation tool
- Dynamically renders Web Controls on screen
- Uses a predefined data structure for this

Questionnaire – Data Structure

- Questionnaire
 - Sections
 - Questions
 - Info (title, type, etc.)
 - ControlSettings
 - Options

Questionnaire - Usage

- Fill using **LoadQuestionnaire**
 - Send filled Data Structure
- Get answers using
 - **WebGet paAnswers**
 - **OnSubmit**

1. Define your questions

```
tQuestionnaire Qal // Questionnaire structs we'll be filling
tQaSection step1 step2 step3
tQaQuestion question1 question2 question3 question4, question5 question6
tControlSettings settings cleanSettings // Settings struct we'll be re
tQaOption opt1 opt2 opt3 // Options for certain questions. We'll be

//===== Questionnaire 2 =====/

// ===== Question 1 (Openvraag-getal)
Move questionOpenNumber to question1.iType
Move 1 to question1.iQuestionId
Move 2 to question1.iNextQuestionId
Move "Speed (in m/s)" to question1.sDescription
Move "120" to question1.sPlaceholder
Move "SPEED" to question1.sCodeExt

Move True to question1.bIsVisible
Move True to question1.bIsMandatory
Move False to question1.bDecision

// Question 2 (Openvraag-getal)
Move questionOpenNumber to question2.iType
Move 2 to question2.iQuestionId
Move 3 to question2.iNextQuestionId
Move "Time (in sec)" to question2.sDescription
Move "5" to question2.sPlaceholder
Move "TIME" to question2.sCodeExt

Move True to question2.bIsVisible
Move True to question2.bIsMandatory
Move False to question2.bDecision
```

2. Add to sections, load Questionnaire

```
// Add questions to sections
Move question1 to section1.aQuestions[0]
Move question2 to section1.aQuestions[1]
Move question3 to section1.aQuestions[2]
Move question4 to section1.aQuestions[3]
Move question5 to section2.aQuestions[0]
Move question6 to section2.aQuestions[1]
Move question7 to section2.aQuestions[2]

// Add the sections to the Questionnaire to load
Move section1 to Qal.aSections[0]
Move section2 to Qal.aSections[1]

// Load structure into QA, send to client
Send LoadQuestionnaire of oQuestionnaire Qal
```

3. Magic!...

Get the answers

```
Object oWebButton3 is a cWebButton
  Set piColumnSpan to 10
  Set psCaption to "Display given answers"

Procedure OnClick
  String sAnswers
  tQaAnswer[] aAnswers
  Integer iArrSize
  Integer i

  WebGet paAnswers of oQuestionnaire to aAnswers
  Move (SizeOfArray(aAnswers)) to iArrSize
  Showln iArrSize
  Move 0 to i

  While (i < iArrSize)
    Move (sAnswers + (String(aAnswers[i].iQuestionId) + ". " + aAnswers[i].sAnswer) + Character(13)) to sAnswers

    Move (i + 1) to i
  Loop

  Send ShowInfoBox sAnswers

End_Procedure
End_Object
```

Questionnaire - Highlights

- Dynamically render controls at runtime
- Fully configurable, like standard web controls
- Decision trees
- Formulas

PdfViewer

- Display PDFs in a uniform way
 - Any browser
 - Any device
- Very relevant for mobile devices
- Uses a WebIframe with ViewerJs library

PdfViewer - Usage

- Very easy to use:
 - Add ViewerJs folder to your AppHtml folder
 - (Web)Set psUrl to “viewerJS/#../” followed by PDF location
 - PDF location can also be obtained using Get DownloadUrl ...

Loading a pdf

```
Object oWebButton1 is a cWebButton
  Set piColumnSpan to 3
  Set psCaption to "Load pdf 1"

  Procedure OnClick
    Send ClearIframe of oWebIFrame1
    WebSet pbRender of oWebImagel to False
    WebSet pbRender of oWebIFrame1 to True
    WebSet psUrl of oWebIFrame1 to "ViewerJS/#../Uploads/Scheduler-information.pdf"
  End_Procedure
End_Object
```

With Get DownloadURL

```
Object oWebButton2 is a cWebButton
  Set piColumnSpan to 3
  Set piColumnIndex to 3
  Set psCaption to "Load pdf 2"

  Procedure OnClick
    String sUrl sFilePath

    Send ClearIframe of oWebIFrame1
    WebSet pbRender of oWebImagel to False
    WebSet pbRender of oWebIFrame1 to True

    Get GetDownloadFolderPath of ghoWebApp to sFilePath
    Move (sFilePath + "Scheduler changelog 2016-06-03.pdf") to sFilePath

    Get DownloadURL of ghoWebResourceManager sFilePath to sUrl

    Move ("ViewerJS/#../" + sUrl) to sUrl
    WebSet psUrl of oWebIFrame1 to sUrl
  End_Procedure
End_Object
```


Image Zoom

- Control to allow for zooming on an image
- Uses a JS + HTML canvas
- Works with both mouse and touch gestures
- Several configuration possibilities and fit modes

Image Zoom - Usage

- Simple usage:
 - Create a WebImageZoom Object
 - (Web)Set psUrl to image location
 - Can also be through Get DownloadUrl

Object declaration

```
Object oWebImageZoom1 is a cWebImageZoom

    Set psUrl to ""
    Set piColumnSpan to 0
    Set pbFillHeight to True
    Set pbShowBorder to True
    Set psToolTip to "Click and move your mouse or scroll to zoom!"
//    Set piHeight to 100

End_Object
```

Loading an image

```
Object oWebButton1 is a cWebButton
    Set piColumnSpan to 12
    Set psCaption to "Load image"
    Set piColumnIndex to 12

    Procedure OnClick
        String sFilePath sUrl

        Get GetDownloadFolderPath of ghoWebApp to sFilePath
        Move (sFilePath + "Ivy.jpg") to sFilePath

        Get DownloadURL of ghoWebResourceManager sFilePath to sUrl

        WebSet psUrl of oWebImageZoom1 to sUrl
    End_Procedure
End_Object
```

Where do i get this!?

- Scheduler:
 - Purchase the DF Wrapper – 750
 - Purchase dHtmlx license based on corporate needs
 - Demo at: schedulerdemo.dataaccess.eu (login U: demo P: demo)
- Questionnaire:
 - Work in progress, will be released soon
- PdfViewer:
 - Demo workspace available on request
- Image Zoom:
 - Demo workspace available on request

Thank you for your attention
Any questions?

